
บทที่   Ã
หลักการพื้นฐานในการเขียนโปรแกรม

จดุประสงคเชิงพฤติกรรม
1. เขาใจพื้นฐานการเขียนโปรแกรม
2. เขาใจการเขียนโปรแกรมเมน ูเพื่อเรียกใชงานโปรแกรมอื่น ๆ ที่พัฒนาขึ้น
3. เขาใจการเชื่อมแฟมขอมูล
4. เขาใจการเขียนโปรแกรมเพื่อปรับปรุงขอมูล เชน เพิ่ม ลบ และแกไข

หวัขอในบทเรียน
4.1  ศึกษาการทํ าซํ้ าดวยพีระมิด
4.2  สูตรคูณ และการคํ านวณ
4.3  การอานขอมูลมาพิมพ
4.4  การแตงจอภาพอยางงาย
4.5  การเขียนเมนู
4.6  การเชื่อมแฟม 2 แฟม
4.7  การสรางเสียง
4.8  โปรแกรมขอสอบ
4.9  การปรับปรุงขอมูล
ตวัอยางโปรแกรมเกี่ยวกับโปรแกรมที่นาสนใจ  69  ตัวอยาง

หมายเหตุ
เนือ้หาในบทนีส้รางขึ้น เพื่อใชเปนแนวในการสอนพื้นฐานการเขียนโปรแกรมอยางงาย ตัวอยางสวนใหญผู

เรยีนสามารถน ําความรูเดิมมาเปนพื้นฐานไดทันที เชน DBASE, C, BASIC และ PASCAL เปนตน สวนขอสอบ 
Online และเมน ูเปนการประยุกตมาจากความตองการของผูเขียน  ส ําหรับค ําสั่ง หรือฟงกชันที่ไมมีการอธิบายไวโดย
ละเอยีด ผูเรียนสามารถเปดดูรายละเอียดจากบทที่ 2 และบทที ่3 เพือ่ท ําความเขาใจไดทันที

บทที่ 4   หลักการพื้นฐานในการเขียนโปรแกรม
ในบทนีจ้ะรวมพืน้ฐานการเขียนโปรแกรมในลักษณะตาง ๆ  เพื่อเปนพื้นฐานที่จะเขาใจการเขียนโปรแกรม

ในหลายแบบ เชน การท ําความเขาใจการทํ าซํ้ าโดยศึกษาจากพีระมิด  การจัดการกับขอมูลโดยเชื่อมขอมูลของแฟม
การปรบัปรุงแฟมขอมูล หรือการอานขอมูลมาทํ ารายงานเปนตน

&  4.1 ศึกษาการทํ าซํ ้าดวยพีระมิด

4.1.1 การพิมพ 1 ถึง 10
ในหวัขอนีจ้ะแสดงใหเห็น 2 คํ าสั่ง เพื่อใชทํ าซํ้ า คือ FOR และ WHILE โดยยกตัวอยางเกี่ยวกับการวนลูป

เพือ่พมิพเลข 1 ถึง 10 เพราะเปนโปรแกรมในรูปแบบนี้เขาใจไดงาย ๆ
: ตัวอยางที่ 4.1
//การพมิพ 1 ถึง 10 บรรทัดละ 1 จํ านวน ดวย FOR แบบธรรมดา
FOR I = 1 TO 10
    ? I
NEXT
// หรอืเขียนอีกแบบหนึ่งไดดังนี ้  FOR I = 1 TO 10 ; ? I ; NEXT
: ตัวอยางที่ 4.2
//การพมิพ 1 ถึง 10 บรรทัดละ 3 จํ านวน ดวย FOR แบบธรรมดา
//โดยแสดงใหเห็นการใชคํ าสั่ง ? และ ??
//ซึง่ ? หมายถึง การปดบรรทัดแลวพิมพ
FOR I = 1 TO 10
    ?? I,I
    ?? I
    ? '====='  ;    ?
NEXT
: ตัวอยางที่ 4.3
//การพมิพ 1 ถึง 10 โดย WHILE โดยกํ าหนดคาเริ่มตนเปน 0
//แตจะแสดง 1 ถึง 10 ทั้งหมด 2 รอบ โดยแยกคํ าสั่ง WHILE และ DO WHILE
I = 0
WHILE I < 10
   I++
   ? I
END

? '---------------------'
I = 0
DO WHILE I < 10
   I = I + 1
   ? I
ENDDO
: ตัวอยางที่ 4.4
//การพมิพ 1 ถึง 10 โดย WHILE โดยกํ าหนดคาเริ่มตนเปน 1
//ตวัอยางนี้จะพิมพ 1 ถึง 10 ทั้งหมด  3 รอบ
I = 1
WHILE I <= 10
   ? I
   I++
END
? '-------------'
I = 1
DO WHILE I <= 10
   ? I
   I = I + 1
ENDDO
? '-------------'
I = 1
DO WHILE I < 11
   ? I
   I++
ENDDO

4.1.2 พิมพชุดของตัวเลขเปนรูปสามเหลี่ยมชิดซาย
ในหวัขอนี ้ตองการใหเรียนรูการใชลูปของ คํ าสั่ง FOR 2 ชุด  โดยคํ าสั่ง FOR ชุดแรกใชคุมจํ านวนแถว และ

ค ําสัง่ FOR ชุดใน ใชคุมหลักของแตละแถว

ต.ย.ผลลัพธ
1
2
3
4
5
6
7
8
9
10
1
2
3
4
5
6
7
8
9
10
1
2
3
4
5
6
7
8
9
10

: ตัวอยางที่ 4.5
// I เปนตวัเลข ใชเนื้อที่ในการพิมพมาก จึงทํ าเปนตัวอักษร แลวตัดชองวางทิ้ง
FOR I = 1 TO 3           // คุมจํ านวนบรรทัด
   FOR J = 1 TO 5        // คมุจ ํานวนตัวเลขที่ใชพิมพในแตละบรรทัด
       ?? LTRIM(STR(I))
   NEXT  // ใชคุมจ ํานวนหลักในแตละบรรทัด
   ?
NEXT  // ใชคุมจํ านวนบรรทัด
: ตัวอยางที่ 4.6
FOR I = 1 TO 3           // คุมจํ านวนบรรทัด
   FOR J = 1 TO I        // คมุจ ํานวนตัวเลขที่ใชพิมพในแตละบรรทัด
       ?? LTRIM(STR(I))
   NEXT
   ?
NEXT
: ตัวอยางที่ 4.7
FOR I = 1 TO 3           // คุมจํ านวนบรรทัด
   FOR J = 1 TO I        // คมุจ ํานวนตัวเลขที่ใชพิมพในแตละบรรทัด
       ?? LTRIM(STR(J))
   NEXT
   ?
NEXT
: ตัวอยางที่ 4.8
FOR I = 1 TO 3             // คุมจํ านวนบรรทัด
   FOR J = I TO 1   STEP -1  // คมุจ ํานวนตัวเลขที่ใชพิมพในแตละบรรทัด
       ?? LTRIM(STR(J))
   NEXT
   ?
NEXT
: ตัวอยางที่ 4.9
FOR I = 1 TO 3
   FOR J = 1 TO (2*I-1)
       ?? LTRIM(STR(J))
   NEXT
   ?
NEXT

ต.ย.ผลลัพธ
11111
22222
33333

ต.ย.ผลลัพธ
1
22
333

ต.ย.ผลลัพธ
1
12
123

ต.ย.ผลลัพธ
1
21
321

ต.ย.ผลลัพธ
1
123
12345

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                    บทที่ 4  :  หลักการพื้นฐานในการเขียนโปรแกรม                                              181

: ตัวอยางที่ 4.10
// ถาใช , กัน้ระหวางตัวแปรในค ําสั่ง ?? จะทํ าใหเกิดชองวาง 1 ชอง
FOR I = 1 TO 3
   ?? '*'
   FOR J = 1 TO (2*I-1)
       ?? LTRIM(STR(J))
   NEXT
   ?? '*',LTRIM(STR(I))
   ?
NEXT

4.1.3 พิมพชุดของตัวเลขเปนรูปสามเหลี่ยมชิดขวา
ในหวัขอนี ้ตองการใหเรียนรูการใชลูปของ FOR 3 ชุด  โดยคํ าสั่ง FOR ชุดแรกใชคุมจํ านวนบรรทัด และจะ

คมุค ําสัง่ FOR อกี 2 ชุด  ในขณะที่คํ าสั่ง FOR ชุดในชุดแรกใชคุมหลักของแตละบรรทัด  เพื่อพิมพชองวางและคํ าสั่ง
FOR ชดุในชดุที่สองใชคุมหลักของแตละบรรทัด เพื่อพิมพตัวเลข
: ตัวอยางที่ 4.11
FOR I = 1 TO 3         // คุมจํ านวนบรรทัด
    FOR J = 1 TO 5-I   // พมิพชองวางหนาตัวเลขของแตละบรรทัด
        ?? ' '
    NEXT
    FOR J = 1 TO I     // พมิพตัวเลขของแตละบรรทัด
        ?? LTRIM(STR(J))
    NEXT  ;      ?
NEXT
: ตัวอยางที่ 4.12
FOR I = 1 TO 3           // คุมจํ านวนบรรทัด
    FOR J = 1 TO 5-I     // พมิพชองวางหนาตัวเลขของแตละบรรทัด
        ?? '  '
    NEXT
    FOR J = 1 TO (2*I-1) // พมิพตัวเลขของแตละบรรทัด
        ?? '*'
    NEXT
    ?
NEXT

ต.ย.ผลลัพธ
*1* 1
*123* 2
*12345* 3

ต.ย.ผลลัพธ
  1
 12
123

ต.ย.ผลลัพธ
  *
 ***
*****

: ตัวอยางที่ 4.13
FOR I = 1 TO 3         // คุมจํ านวนบรรทัด
    FOR J = 1 TO 5-I   // พมิพชองวางหนาตัวเลขของแตละบรรทัด
        ?? ' '
    NEXT
    ?? '*'
    FOR J = 1 TO I     // พมิพตัวเลขของแตละบรรทัด
        ?? LTRIM(STR(J))
    NEXT
    ?? '*'
    ?? LTRIM(STR(I))
    ?? '*'
    ?? LTRIM(STR(I*2))
    ?
NEXT

4.1.4 พิมพชุดของตัวเลขเปนรูปสามเหลี่ยมตาง ๆ
ในหวัขอนี ้ตองการใหเรียนรูการใชลูปของ FOR มากกวา 3 ชุดขึ้นไป

: ตัวอยางที่ 4.14
FOR I = 1 TO 3         // คุมจํ านวนบรรทัด
    FOR J = 1 TO 5-I   // พมิพชองวางหนาตัวเลขของแตละบรรทัด
        ?? ' '
    NEXT
    FOR J = 1 TO I     // พมิพตัวเลขของแตละบรรทัดชุดแรก 123
        ?? LTRIM(STR(J))
    NEXT
    FOR J = 1 TO 5     // พมิพตัวเลขของแตละบรรทัดชุดที่สอง 33333
        ?? LTRIM(STR(I))
    NEXT
    ?
NEXT

ต.ย.ผลลัพธ
     *1*1*2
   *12*2*4
 *123*3*6

ต.ย.ผลลัพธ
     111111
   1222222
 12333333

: ตัวอยางที่ 4.15
FOR I = 1 TO 4
    FOR J = 1 TO 5-I
        ?? ' '
    NEXT
    FOR J = 1 TO I
        ?? LTRIM(STR(J))
    NEXT
    FOR J = I-1 TO 1 STEP -1
        ?? LTRIM(STR(J))
    NEXT
    ?
NEXT
: ตัวอยางที่ 4.16
FOR I = 1 TO 3
    FOR J = 1 TO 4-I
        ?? LTRIM(STR(J))
    NEXT
    FOR J = 1 TO I-1
        ?? ' '
    NEXT
    ?? '*'
    FOR J = 1 TO 3-I
        ?? ' '
    NEXT
    FOR J = 1 TO I
        ?? LTRIM(STR(J))
    NEXT
    ?
NEXT

ต.ย.ผลลัพธ
     1
   121
 12321
1234321

ต.ย.ผลลัพธ
123*  1
12 * 12
1  *123

: ตัวอยางที่ 4.17
FOR I = 1 TO 4
    FOR J = 1 TO I
        FOR K = 1 TO J
            ?? LTRIM(STR(K))
        NEXT
    NEXT
    ?
NEXT
: ตัวอยางที่ 4.18
CLS
LPOS = 10
CPOS = 40
ACCEPT TO X
X = VAL(X)
FOR I = 1 TO X
    Y = MOD(I,4)
    DO CASE
       CASE Y = 1       // ส ําหรับแนวตั้งชิดซาย
            FOR J = 1 TO I
                SETPOS(LPOS+J,CPOS) ; ?? '*'
            NEXT
            SETPOS(LPOS+J,CPOS-1)
       CASE Y = 2       // ส ําหรับแนวนอนเสนลาง
            FOR J = 1 TO I
                SETPOS(LPOS,CPOS+J) ; ?? '*'
            NEXT
            SETPOS(LPOS+1,CPOS+J)
       CASE Y = 3       // ส ําหรับแนวตั้งชิดขวา
            FOR J = 1 TO I
                SETPOS(LPOS-J,CPOS) ; ?? '*'
            NEXT
            SETPOS(LPOS-J,CPOS+1)

ต.ย.ผลลัพธ
1
112
112123
1121231234

ต.ย.ผลลัพธ
IF X=3

     *
 *  *
 ***

ต.ย.ผลลัพธ
IF X=5
 *****
*      *
*  *  *
*  ***
*

       CASE Y = 0       // ส ําหรับแนวนอนเสนบน
            FOR J = 1 TO I
                SETPOS(LPOS,CPOS-J) ; ?? '*'
            NEXT
            SETPOS(LPOS-1,CPOS-J)
    ENDCASE
    LPOS = ROW()
    CPOS = COL()
NEXT

&  4.2 สูตรคูณ และการคํ านวณ
โปรแกรมเกีย่วกับสูตรคูณ เปนโปรแกรมที่นาศึกษา เพราะทํ าความเขาใจไดโดยงาย  และทบทวนการท ําซํ้ า

โดยค ําสัง่วนลปูไดเปนอยางด ี และศึกษาการคํ านวณดวยเครื่องหมายตาง ๆ

4.2.1 สูตรคูณในหลายลักษณะ
ในหวัขอนี ้ตองการใหเขาใจการคํ านวณอยางงาย ๆ โดยศึกษาจากสูตรคูณ

: ตัวอยางที่ 4.19
// พมิพสูตรคูณแม 2 ธรรมดา
CLS
X = 2
FOR I = 1 TO 12
    ? X," * ",I," = ",X*I
NEXT
@ 0,1 TO 13,40    // ตีเสนลอมรอบสูตรคูณ
: ตัวอยางที่ 4.20
// พมิพสูตรคูณแม 2 และ 3 ใหติดกัน
// หมายความวาสูตรคูณแม 2 อยูซีกซาย และสูตรคูณแม 3 อยูซีกขวา
// โดย STR(2,3) มีหนาที่ท ําใหเลข 2 เปนตัวอักษรขนาด 3 ชอง ถูกใชเพื่อความสวยงาม
CLS
FOR I = 1 TO 12
    ?  "2 * ",STR(I,2)," = ",STR(2*I,3)
    ?? " | "
    ?? "3 * ",STR(I,2)," = ",STR(3*I,3)
NEXT

: ตัวอยางที่ 4.21
// พมิพสูตรคูณใหสวยงาม โดยเลือกสูตรคูณจากแปนพิมพได
CLS
SET DELIMITERS TO "[ ]"
SET DELIMITERS ON
X = 2
@ 2,5 SAY "แมสูตรคูณที่ตองการคือ " GET X
READ
SET COLOR TO "B/W"
@ 5,10 CLEAR TO 16,40
FOR I = 1 TO 6
    SETPOS(4+I,15)
    ?? STR(X,2)," * ",STR(I,2)," = ",STR(X*I,3)
NEXT
FOR I = 7 TO 12    // จะใหผลตางกันเพราะ , ใน ?? จะเพิ่มชองวางอีก 1 ชอง
    @ 4+I,15 SAY STR(X,2)+" * "+STR(I,2)+" = "+STR(X*I,3)
NEXT

4.2.2 เครื่องคิดเลข
ในหวัขอนี ้ตองการใหเห็นการรับคาจากแปนพิมพมาคํ านวณ เชนเครื่องคิดเลข

: ตัวอยางที่ 4.22
// รบัคาจากแปนพิมพมา 2 คา แลวหาผลบวก ลบ คูณ หาร และยกกํ าลัง
CLS
X = 0
Y = 0
@ 5,10 SAY "GET  FIRST NUMBER " GET X
@ 6,10 SAY "GET SECOND NUMBER " GET Y
READ
@ 7,10 SAY "RESULT OF + " + STR(X+Y)
@ 8,10 SAY "RESULT OF -" + STR(X-Y)
@ 9,10 SAY "RESULT OF *" + STR(X*Y)
@ 10,10 SAY "RESULT OF /" + STR(X/Y)
@ 11,10 SAY "RESULT OF ^" + STR(X**Y)

: ตัวอยางที่ 4.23
// รบัคา 2 คา และรับตัวดํ าเนินการเปนตัวที่ 3
// โดยใชคํ าสั่ง CASE เลือกคํ านวณ
DO WHILE .T.
   ? "PRESS 0 ON GET SIGN TO EXIT"
   ? "---------------------------"
   ACCEPT "   GET X : " TO X
   ACCEPT "   GET Y : " TO Y
   ACCEPT "GET SIGN : " TO S
   DO CASE
      CASE  S = "+"   ;  ? "YOUR RESULT IS ",VAL(X) + VAL(Y)
      CASE  S = "-"   ;  ? "YOUR RESULT IS ",VAL(X) - VAL(Y)
      CASE  S = "*"   ;  ? "YOUR RESULT IS ",VAL(X) * VAL(Y)
      CASE  S = "/"   ;  ? "YOUR RESULT IS ",VAL(X) / VAL(Y)
      CASE  S = "^"   ;  ? "YOUR RESULT IS ",VAL(X) ^ VAL(Y)
      CASE  S = "0"   ;  EXIT
      OTHERWISE       ;  ? "YOUR SIGN IS ERROR."
   ENDCASE
ENDDO
: ตัวอยางที่ 4.24
// รบัคา 2 คา และรับตัวดํ าเนินการเปนตัวที่ 3
// โดยใชค ําสั่ง CASE เลือกคํ านวณ และสามารถกดปุม ESC เพื่อออกจากโปรแกรม
CLS
DO WHILE LASTKEY() != 27
   X := Y := R := 0
   Z = " "
   @ 5,10 SAY "PRESS ESC TO EXIT"
   @ 6,10 SAY "-----------------"
   @ 7,10 SAY "GET    X : " GET X
   @ 8,10 SAY "GET    Y : " GET Y
   @ 9,10 SAY "GET SIGN : " GET Z VALID Z $ '+-*/^'
   READ

   DO CASE
      CASE  Z = "+"   ;  R = X + Y
      CASE  Z = "-"   ;  R = X - Y
      CASE  Z = "*"   ;  R = X * Y
      CASE  Z = "/"   ;  R = X / Y
      CASE  Z = "^"   ;  R = X ^ Y
   ENDCASE
   IF Z $ '+-*/^'
      @ 10,10 SAY "RESULT IS : " + STR(R)
      INKEY(3)
      @ 10,10 CLEAR
   ENDIF
ENDDO

&  4.3 การอานขอมูลมาพิมพ
การอานขอมูลจากแฟมเดียวมาพิมพ

ใหแฟมนี้ชื่อแฟมพนักงาน   EMPL.DBF
1. รหัสพนักงาน EMPL      9(7)
2. ชือ่พนักงาน NAME      X(20)
3. สกุลพนักงาน SURN      X(20)

4.3.1 การอานขอมูลมาพิมพทางจอภาพ
หวัขอนี ้จะน ําขอมูลจากแฟมมาพิมพ โดยแสดงใหเห็นการวนลูปเพื่อนํ าขอมูลมาแสดงผล

: ตัวอยางที่ 4.25
// อานขอมูลมาพิมพอยางงาย ๆ
USE EMPL
WHILE !EOF()
    ? EMPL,NAME,SURN
    SKIP
END

: ตัวอยางที่ 4.26
// อานขอมูลมาพิมพ และมีล ําดับ โดยใชการวนลูปของ WHILE
USE EMPL
I = 0
DO WHILE !EOF()
   I++
   ? I,EMPL,NAME,SURN
   SKIP
ENDDO
: ตัวอยางที่ 4.27
// อานขอมูลมาพิมพ และมีล ําดับ โดยใชการวนลูปของ FOR
USE EMPL
FOR I = 1 TO RECCOUNT()
    ? I,EMPL,NAME,SURN
    SKIP
NEXT
: ตัวอยางที่ 4.28
// อานขอมูลมาพิมพ แลวหยุดดูทีละหนา หนาละ 10 เรคอรด
USE EMPL
R = 7
WHILE !EOF()
     R++
     @ R,10 SAY STR(EMPL)+NAME+SURN
     SKIP
     IF R = 17
        INKEY(0)
        CLS
        R = 7
     ENDIF
END

: ตัวอยางที่ 4.29
// อานขอมูลมาพิมพบนพื้นส ีที่มีเงา
SET COLOR TO "/W"
CLS
SET COLOR TO "/N"
@ 5,10 CLEAR TO 22,70
SET COLOR TO "W/B"
@ 4,8 CLEAR TO 21,68
USE EMPL
I = 0
R = 6
WHILE !EOF()
     I++  ;    R++
     @ R,10 SAY STR(I)+STR(EMPL)+NAME+SURN
     SKIP
     IF R = 19
        INKEY(0)
        @ 4,8 CLEAR TO 21,68
        R = 6
     ENDIF
END

4.3.2 การอานขอมูลมาพิมพทางเครื่องพิมพ
ในหวัขอนี ้จะแสดงใหเห็นการพิมพขอมูลออกทางเครื่องพิมพ  ในการนํ าขอมูลออกเครื่องพิมพ ควรใชคํ าสง

SAY เพราะควบคุมจุดที่จะพิมพไดแนนอน ตางกับการใช ? คูกับ SET PRINT ON
: ตัวอยางที่ 4.30
// อานขอมูลมาพิมพ ออกทั้งจอภาพและเครื่องพิมพ
USE EMPL
SET PRINT ON
WHILE !EOF()
    ? EMPL,NAME,SURN
    SKIP
END
SET PRINT OFF

: ตัวอยางที่ 4.31
// อานขอมูลมาพิมพ ออกทางเครื่องพิมพเทานั้น
USE EMPL
SET DEVICE TO PRINTER
I = 0
WHILE !EOF()
    I++
    @ I,10 SAY STR(EMPL)+NAME+SURN
    SKIP
END
EJECT
SET DEVICE TO SCREEN
: ตัวอยางที่ 4.32
// อานขอมูลมาพิมพ ออกทางเครื่องพิมพ หรือจอภาพ โดยเลือกจากตัวเลือก
// โดยควบคมุการแสดงผลใหแยกกันไดอยางชัดเจน และถูกตอง
CLS
OPT = ALERT("เลอืกผลลัพธของรายงาน",{"จอภาพ","เครื่องพิมพ","ยกเลิก"})
USE EMPL
IF OPT = 1
   SET COLOR TO "/W"
   CLS
   SET COLOR TO "/N"
   @ 5,10 CLEAR TO 22,70
   SET COLOR TO "W/B"
   @ 4,8 CLEAR TO 21,68
   PG = 15
   SETR  = 6
ELSE
   PG = 30
   SETR = 1
   SET DEVICE TO PRINTER
ENDIF
R = SETR

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


192                                           การเขียนโปรแกรมส ําหรับฐานขอมูลดวย Clipper                                                         

I = 0
WHILE !EOF() .AND. OPT != 3
     I++
     R++
     @ R,10 SAY STR(I)+STR(EMPL)+NAME+SURN
     SKIP
     IF R = PG
        IF OPT = 1
           INKEY(0)
           @ 4,8 CLEAR TO 21,68
        ELSE
           EJECT
        ENDIF
        R = SETR
     ENDIF
END
SET DEVICE TO SCREEN

4.3.3 การอานขอมูลมาค ํานวณแลวพิมพ
หวัขอนี ้ไดน ําแฟมมาคํ านวณเพียงแฟมเดียว เพื่อหาคาตาง ๆ เชน ภาษี โบนัส รายไดจริง เปนตน
แฟมเงินเดือน SALARY.DBF
1. รหัสพนักงาน EMPL     9(7)
2. เงนิเดือนพนักงาน SALARY 9(7)V99

: ตัวอยางที่ 4.33
// ค ํานวณภาษ ี7 % และรายไดที่ตองรับจริงของพนักงาน
USE SALARY
WHILE !EOF()
    TAX = SALARY * 0.07
    INCOME = SALARY - TAX
    ? EMPL,SALARY,TAX,INCOME
    SKIP
END

: ตัวอยางที่ 4.34
// ค ํานวณผลรวมของเงินเดือน และผลรวมของรายได
// สิง่ทีต่องสงัเกตในตัวอยางนี้คือ การคํ านวณผลรวมจะคํ านวณในลูป
// แตการพิมพผลรวมตองพิมพหลังจากที่อานขอมูลจนหมดกอน
USE SALARY
TINCOME = 0    // ตองก ําหนด เพราะมีการเรียกใชตอนหาผลรวม
TSALARY = 0
WHILE !EOF()
    TAX = SALARY * 0.07
    INCOME = SALARY - TAX
    TINCOME = TINCOME + INCOME
    TSALARY = TSALARY + SALARY
    ? EMPL,SALARY,TAX,INCOME
    SKIP
END
? "TINCOME = ",TINCOME
? "TSALARY = ",TSALARY

&  4.4 การแตงจอภาพอยางงาย
ความสวยงาม หรอืสีสัน จํ าเปนสํ าหรับการเขียนโปรแกรมใหญ ๆ เพื่อแยกความแตกตาง หรือชวยใหเขาใจ

ไดงายขึน้ในรายละเอียดบางอยาง  จึงควรเขาใจการใชเสน การตีกรอบ การระบายสี สีพื้น สีเงา วันที่ และเวลา

4.4.1 การตีกรอบ สีพื้น และสีเงา โดยท ําเปนนาฬิกา
ในหวัขอนี ้จะทํ านาฬิกาเดินบนพื้นสีตามวินาที

: ตัวอยางที่ 4.35
// นาฬิกาบนพื้นส ีเลิกท ํางานเมื่อกดปุม ESC
SET COLOR TO "N/W"
CLS                      // ระบายสีพื้น
@ 0,0 TO 24,79           // ตีกรอบนอก
SET COLOR TO "/N"
@ 5,7 CLEAR TO 23,74     // ท ําเงากอน
SET COLOR TO "GR+/B"
@ 4,5 CLEAR TO 22,72     // ท ําพื้นสีเปนฉากหลัง

@ 1,1 SAY PADC("ABC COMPANY",78)
@ 2,1 TO 2,78            // ขีดเสนใตชื่อบริษัท
SET CURSOR OFF          // ท ําใหตัวกระพริบ(CURSOR) หายไป
WHILE LASTKEY() != 27
   SETPOS(10,35)
   ?? TIME()             // นาฬกิาจะเดินและเปลี่ยนเลขทุกวินาที
   INKEY(1)
END

4.4.2 การน ําฟงกชันตาง ๆ มาแสดงบนจอภาพ
หวัขอนี ้จะนํ าฟงกชันที่นาสนใจมาแสดงผลบนจอภาพ

: ตัวอยางที่ 4.36
// โปรแกรมนี้มีฟงกชันตาง ๆ ที่นาสนใจหลายตัว
// โดยการระบายสียังคงใชหลักการเดิม
SET COLOR TO W/B
CLS                      // ระบายสีพื้น
@ 0,0 TO 24,79 DOUBLE    // ตีกรอบนอก เสนคู
SET COLOR TO /N
@ 5,7 CLEAR TO 23,74            // ท ําเงากอน
SET COLOR TO GR+/BR
@ 4,5,22,72 BOX '........ '     // ท ําพื้นสีเปนฉากหลัง
@ 1,1 SAY PADC("CDF COMPANY",78)
@ 2,1 TO 2,78                   // ขีดเสนใตชื่อบริษัท
SET CURSOR OFF                  // ท ําใหตัวกระพริบ(CURSOR) หายไป
@ 10,35 SAY "DATE    : "+DTOC(DATE())
@ 11,35 SAY "TIME    : "+TIME()
@ 12,35 SAY "SECOND  : "+STR(SECOND())
@ 13,35 SAY "OS      : "+OS()
@ 14,35 SAY "MEMORY  : "+STR(MEMORY())
@ 15,35 SAY "VERSION : "+VERSION()
@ 16,35 SAY "PRINTER : "       // ถาไมเปดเครื่องพิมพ คาที่ไดจะเห็น .F.
?? ISPRINTER()                 // ไมสามารถแปลงดวยฟงกชัน STR หรือ VAL ได

&  4.5 การเขียนเมนู
การเขยีนเมนมูหีลายวิธ ีทั้งที่ใชฟงกชันของเครื่อง หรือควบคุมการกดแปนพิมพเอง

4.5.1 เมนูมิติเดียว
หวัขอนี ้จะแสดงใหเห็นการเขียนเมนูแบบตาง ๆ ทั้ง GET , PROMPT และ ACHOICE

: ตัวอยางที่ 4.37
// เมนนูีใ้ชการพิมพเมนูดวยค ําสั่ง TEXT และรับคาดวย ACCEPT
CLS
OPT = 0
DO WHILE .T.
   TEXT
     1. พิมพ 1 ถึง 10
     2. พิมพ สูตรคูณแม 2
     3. เลิกการทํ างาน
   ENDTEXT
   ACCEPT TO OPT
   DO CASE
      CASE OPT = "1" ; FOR I = 1 TO 10 ; ? I ; NEXT
      CASE OPT = "2" ; FOR I = 1 TO 12 ; ? I," * 2 = ",I*2 ; NEXT
      CASE OPT = "3" ; QUIT
   ENDCASE
   INKEY(2)
ENDDO
: ตัวอยางที่ 4.38
// เมนนูี้ใช WAIT รับคา และใช IF เลือกท ํางาน
CLS
OPT = 0
DO WHILE OPT != 51
   ? "1. พิมพ 1 ถึง 10"
   ? "2. พิมพ สูตรคูณแม 2"
   ? "3. เลิกการท ํางาน"
   WAIT "รอรับ ...ตัวเลือก"   // กด 1 จะไดเลข 49 เก็บใน OPT
   OPT = LASTKEY()
   IF OPT = 49 ; FOR I = 1 TO 10 ; ? I ; NEXT ; ENDIF

   IF OPT = 50 ; FOR I = 1 TO 12 ; ? I," * 2 = ",I*2 ; NEXT ; ENDIF
   INKEY(2)
ENDDO
: ตัวอยางที่ 4.39
// โปรแกรมจะวิ่งไปตลอดเวลา แตจะทํ างานเมื่อมีการกดตัวเลือก
// จดุดขีองโปรแกรมนี้คือ นาฬิกาจะเดินขณะที่ทํ างานอยู
CLS
OPT = 0
DO WHILE .T.
   @ 4,5 SAY TIME()
   @ 5,5 SAY "1. พิมพ 1 ถึง 10"
   @ 6,5 SAY "2. พิมพ สูตรคูณแม 2"
   @ 7,5 SAY "3. เลิกการท ํางาน หรือปุม ESC"
   // กด 1 จะไดเลข 49 เก็บใน OPT
   INKEY(1)
   IF LASTKEY() != 0
     OPT = LASTKEY()
     IF OPT = 49 ; FOR I = 1 TO 10 ; ? I ; NEXT ; ENDIF
     IF OPT = 50 ; FOR I = 1 TO 12 ; ? I," * 2 = ",I*2 ; NEXT ; ENDIF
     IF OPT = 51 .OR. OPT = 27 ; QUIT ; ENDIF
     INKEY(3)
     KEYBOARD CHR(0)      // สงคาเขาหนวยความจํ าของแปนพิมพ
     INKEY()              // เพือ่ลางคาใน LASTKEY()
     CLS                  // ถาไมลางจะท ําใหคาใน LASTKEY() เทาเดิมตลอดเวลา
   ENDIF
   OPT = 0
ENDDO
: ตัวอยางที่ 4.40
// เลอืกตัวเลือกแบบดึงลง (PULLDOWN MENU)
// ดวยคํ าสั่ง PROMPT
CLS
OPT = 0
DO WHILE LASTKEY() != 27
   @ 4,5 SAY TIME()
   @ 5,5 PROMPT [1. พมิพ 1 ถึง 10]
   @ 6,5 PROMPT [2. พมิพ สูตรคูณแม 2]
   @ 7,5 PROMPT [3. เลกิการท ํางาน หรือปุม ESC]

   MENU TO OPT
   DO CASE
      CASE OPT = 1 ; FOR I = 1 TO 10 ; ? I ; NEXT
      CASE OPT = 2 ; FOR I = 1 TO 12 ; ? I," * 2 = ",I*2 ; NEXT
      CASE OPT = 3 ; QUIT
   ENDCASE
   INKEY(2)
ENDDO
: ตัวอยางที่ 4.41
// เลอืกตัวเลือกแบบดึงลง (PULLDOWN MENU)
// ดวยคํ าสั่ง ACHOICE
CLS
OPT = 0
CHOICE = {"1. พิมพ 1 ถึง 10","2. พิมพ สูตรคูณแม 2",;
          "3. เลิกการท ํางาน หรือปุม ESC"}
DO WHILE LASTKEY() != 27
   OPT = ACHOICE (5,5,7,40,CHOICE)
   DO CASE
      CASE OPT = 1 ; FOR I = 1 TO 10 ; ? I ; NEXT
      CASE OPT = 2 ; FOR I = 1 TO 12 ; ? I," * 2 = ",I*2 ; NEXT
      CASE OPT = 3 ; QUIT
   ENDCASE
   INKEY(2)
ENDDO
: ตัวอยางที่ 4.42
// เลอืกตวัเลอืกโดยควบคุมดวยค ําสั่งรับคา LASTKEY()
// นาฬิกาเดินไดตลอดเวลา
SET COLOR TO W/B
CLS
SET CURSOR OFF
OPT = 1
ARCHOICE = ARRAY(3)
ARCHOICE[1] = [1. พมิพ 1 ถึง 10]
ARCHOICE[2] = [2. พมิพ สูตรคูณแม 2]
ARCHOICE[3] = [3. เลกิการท ํางาน หรือปุม ESC]

DO WRITEMENU                            // พิมพเมนูครั้งแรก
DO WHILE LASTKEY() != 27
   @ 3,5 SAY TIME()                     // แสดงเวลาตลอดเวลา
   INKEY(0.2)                           // หยุดรับแปนพิมพทุก 0.2 วินาที
   IF LASTKEY() != 0
      DO CASE
         CASE LASTKEY() = 24     // กดลง
              IF OPT = 3         // ถาถงึตัวลางสุดแลวยังกดลง
                 OPT = 1         // ให OPT ไปที่ตัวแรก
              ELSE
                 OPT = OPT + 1
              ENDIF
         CASE LASTKEY() = 5      // กดขึ้น
              IF OPT = 1        // ถาถึงตัวแรกแลวยังกดขึ้น
                 OPT = 3         // ให OPT ไปยังตัวสุดทาย
              ELSE
                 OPT = OPT - 1
              ENDIF
         CASE LASTKEY() = 13     // กดเอ็นเทอร (ENTER)
              DO CASE
                 CASE OPT <= 2
                      @ 20,5 SAY "YOUR CHOICE IS "+STR(OPT)
                 CASE OPT = 3
                      @ 20,5 SAY "BYE BYE BYE"
                      EXIT
              ENDCASE
         ENDCASE
         DO WRITEMENU
         KEYBOARD CHR(0)
         INKEY()                // เพือ่ตัง้คาของ LASTKEY() ใหม
     ENDIF
ENDDO
PROCEDURE WRITEMENU
  SET COLOR TO W/B
  I = 7                             // เมนูเริ่มพิมพบรรทัดที ่7 หลักที ่10
  FOR J = I TO I+2                  // เลข 2 หมายถึงจํ านวนตัวเลือกมี 3 ตัว
      @ J,10 SAY ARCHOICE[J-I+1]    // พมิพเมนูทั้งหมดบนจอ
  NEXT
  SET COLOR TO B/W
  @ OPT+I-1,10 SAY ARCHOICE[OPT]   // พมิพตัวเลือกปจจุบัน
RETURN

4.5.2 เมนูหลายมิติ
หวัขอนี ้จะแสดงการสรางเมนูยอยของตัวเลือกในเมนูหลักใหเห็นอยางชัดเจน  ท ําใหผูอานสามารถเขาใจ

และน ําไปพัฒนาโปรแกรมที่เปนระบบ เปนไปไดยิ่งขึ้น
: ตัวอยางที่ 4.43
// เมนหูลายมิต ิโดยเลือกตัวเลือกดวยคํ าสั่ง PROMPT
// ซึง่เมนยูอย จะเรียงไปดานขาง ตามหลักที่ 10, 30 และ 50
// สามารถกดตัวอักษรตัวแรกของตัวเลือก เพื่อเลือกได
SET CURSOR OFF
ARMAIN = 3
ARCM = ARRAY(ARMAIN)
ARCM[1] = [1. CHOICE1]
ARCM[2] = [2. CHOICE2]
ARCM[3] = [3. EXIT]
ARS1 = 2
ARC1 = ARRAY(ARS1)
ARC1[1] = [1. CHOICE11]
ARC1[2] = [2. EXIT TO MAIN]
ARS2 = 4
ARC2 = ARRAY(ARS2)
ARC2[1] = [1. CHOICE21]
ARC2[2] = [2. CHOICE22]
ARC2[3] = [3. CHOICE23]
ARC2[4] = [4. EXIT TO MAIN]
OPT = 1
DO WHILE LASTKEY() != 27
   SET COLOR TO W/B
   CLS
   DO CASE
      CASE OPT >= 1  .AND. OPT <= 3
          @ 7,10 PROMPT ARCM[1]            // พิมพเมนูบนจอภาพ
          @ 8,10 PROMPT ARCM[2]
          @ 9,10 PROMPT ARCM[3]
          MENU TO OPT

      CASE OPT >= 11 .AND. OPT <= 12
          @ 7,30 PROMPT ARC1[1]
          @ 8,30 PROMPT ARC1[2]
          MENU TO OPT
          OPT = OPT + 10
      CASE OPT >= 21 .AND. OPT <= 24
          @ 7,50 PROMPT ARC2[1]           // พิมพเมนูบนจอภาพ
          @ 8,50 PROMPT ARC2[2]
          @ 9,50 PROMPT ARC2[3]
          @ 10,50 PROMPT ARC2[4]
          MENU TO OPT
          OPT = OPT + 20
   ENDCASE
   // =====================================
   DO CASE
      CASE OPT = 1
             OPT = 11
      CASE OPT = 2
             OPT = 21
      CASE OPT = 3
             @ 20,5 SAY "BYE BYE BYE"
             EXIT
      CASE OPT=12 .OR. OPT=24                    // กลับเมนูหลัก
             OPT = 1
      CASE (OPT>=11 .AND. OPT<=12) .OR.;
           (OPT>=21 .AND. OPT<=24)               // กระท ําตัวเลือก
             @ 23,5 SAY "OPT "+STR(OPT)
             INKEY(1)
    ENDCASE
ENDDO

: ตัวอยางที่ 4.44
// เมนหูลายมติ ิโดยเลือกตัวเลือกดวยคํ าสั่งตรวจสอบแปนพิมพ (LASTKEY)
// เมนนูีน้าฬิกาจะเดินตลอดเวลา  แตไมตรวจสอบปุมลูกศรขึ้นหรือลง
SET CURSOR OFF
ARMAIN = 3
ARCM = ARRAY(ARMAIN)
ARCM[1] = [1. CHOICE1]
ARCM[2] = [2. CHOICE2]
ARCM[3] = [3. EXIT]
ARS1 = 2
ARC1 = ARRAY(ARS1)
ARC1[1] = [1. CHOICE11]
ARC1[2] = [2. EXIT TO MAIN]
ARS2 = 4
ARC2 = ARRAY(ARS2)
ARC2[1] = [1. CHOICE21]
ARC2[2] = [2. CHOICE22]
ARC2[3] = [3. CHOICE23]
ARC2[4] = [4. EXIT TO MAIN]
OPT = 1                                 // ก ําหนดตัวเลือกเริ่มตน
TOPLMT = OPT                            // ก ําหนดบรรทัดบนสุดของเมนูปจจุบัน
BOTTOMLMT = ARMAIN                      // ก ําหนดบรรทัดลางสุดของเมนูปจจุบัน
DO WRITEMENU                            // พิมพเมนูครั้งแรก
DO WHILE LASTKEY() != 27
   @ 3,5 SAY TIME()                     // แสดงเวลาตลอดเวลา
   INKEY(0.2)                           // หยุดรับแปนพิมพทุก 0.2 วินาที
   IF LASTKEY() != 0
      DO CASE
         CASE LASTKEY() = 24     // กดลง
              IF OPT = BOTTOMLMT // ถาถงึตัวลางสุดแลวยังกดลง
                 OPT = TOPLMT   // ให OPT ไปที่ตัวแรก
              ELSE
                 OPT = OPT + 1

              ENDIF
         CASE LASTKEY() = 5      // กดขึ้น
              IF OPT = TOPLMT    // ถาถึงตัวแรกแลวยังกดขึ้น
                 OPT = BOTTOMLMT // ให OPT ไปยังตัวสุดทาย
              ELSE
                 OPT = OPT - 1
              ENDIF
         CASE LASTKEY() = 13     // กดเอ็นเทอร (ENTER)
              DO CASE
                 CASE OPT = 1
                      OPT = 11   ;    TOPLMT = OPT
                      BOTTOMLMT = ARS1 + OPT - 1
                 CASE OPT = 2
                      OPT = 21   ;    TOPLMT = OPT
                      BOTTOMLMT = ARS2 + OPT - 1
                 CASE OPT = 3
                      @ 20,5 SAY "BYE BYE BYE"
                      EXIT
                 CASE OPT=10+ARS1 .OR. OPT=20+ARS2  // กลับเมนูหลัก
                      OPT = 1
                      TOPLMT = OPT
                      BOTTOMLMT = ARMAIN
                 CASE OPT >= 11 .AND. OPT<=10+ARS1  // กระท ําตัวเลือก
                      @ 20,5 SAY "OPT "+STR(OPT)
                      INKEY(3)
                 CASE OPT >= 21 .AND. OPT<=20+ARS2  // กระท ําตัวเลือก
                      @ 20,5 SAY "OPT "+STR(OPT)
                      INKEY(3)
              ENDCASE
         ENDCASE
         DO WRITEMENU
         KEYBOARD CHR(0)
         INKEY()                // เพือ่ตัง้คาของ LASTKEY() ใหม

     ENDIF
ENDDO
PROCEDURE WRITEMENU
  SET COLOR TO W/B
  CLS
  DO CASE
     CASE OPT >= 1  .AND. OPT <= ARMAIN
          I = 7                         // เมนูเริ่มพิมพบรรทัดที ่7
          FOR J = I TO I+ARMAIN-1
              @ J,10 SAY ARCM[J-I+1]    // พิมพเมนูบนจอภาพ
          NEXT
          TEMPAR = ARCM
     CASE OPT >= 11 .AND. OPT <= ARS1 + 10
          I = 12                        // เมนูเริ่มพิมพบรรทัดที ่7
          FOR J = I TO I+ARS1-1
              @ J,10 SAY ARC1[J-I+1]    // พิมพเมนูบนจอภาพ
          NEXT
          TEMPAR = ARC1
     CASE OPT >= 21 .AND. OPT <= ARS2 + 20
          I = 18                        // เมนูเริ่มพิมพบรรทัดที ่18
          FOR J = I TO I+ARS2-1
              @ J,10 SAY ARC2[J-I+1]    // พิมพเมนูบนจอภาพ
          NEXT
          TEMPAR = ARC2
  ENDCASE
  SET COLOR TO B/W
  @ I+MOD(OPT,10)-1,10 SAY TEMPAR[MOD(OPT,10)]    // พมิพตัวเลือกปจจุบัน
RETURN

: ตัวอยางที่ 4.45
// ตวัอยางนี ้ไดรวมเอาความสามารถตาง ๆ ของเมนูไว
// สามารถตรวจสอบ การกดปุมซาย ขวา บน ลาง
// สามารถ กดปุม ESC เพื่อเลิกการทํ างานไดทันที
// ยงัคงท ําใหนาฬิกาเดินไดตลอดเวลา
SET CURSOR OFF
ARMAIN = 3  ;   ARCM = ARRAY(ARMAIN)
ARCM[1] = [1. CHOICE1]
ARCM[2] = [2. CHOICE2]
ARCM[3] = [3. EXIT]
ARS1 = 2  ;   ARC1 = ARRAY(ARS1)
ARC1[1] = [1. CHOICE11]
ARC1[2] = [2. EXIT TO MAIN]
ARS2 = 4    ;   ARC2 = ARRAY(ARS2)
ARC2[1] = [1. CHOICE21]
ARC2[2] = [2. CHOICE22]
ARC2[3] = [3. CHOICE23]
ARC2[4] = [4. EXIT TO MAIN]
OPT = 1                                 // ก ําหนดตัวเลือกเริ่มตน
TOPLMT = OPT                            // ก ําหนดบรรทัดบนสุดของเมนูปจจุบัน
BOTTOMLMT = ARMAIN                      // ก ําหนดบรรทัดลางสุดของเมนูปจจุบัน
ALLSUB = 2                              // จ ํานวนเมนูยอยทั้งหมด
DO WRITEMENU                            // พิมพเมนูครั้งแรก
DO WHILE LASTKEY() != 27
   @ 1,5 SAY TIME()                     // แสดงเวลาตลอดเวลา
   INKEY(0.2)                           // หยุดรับแปนพิมพทุก 0.2 วินาที
   IF LASTKEY() != 0
      // กรณกีดปุมใด ๆ จะมาตรวจสอบในสวนนี้
      DO CASE
         CASE LASTKEY() = 27
              EXIT
         CASE LASTKEY() = 4      // กดขวา
              IF OPT < 10                    // ตวัเลือกในเมนูหลัก

                 DO PRESSDOWN
              ELSE
                 IF (OPT-MOD(OPT,10))/10 = ALLSUB // ขามไปเมนูทางขวาหรือเริ่มใหม
                    OPT = 11
                 ELSE
                    OPT = (OPT-MOD(OPT,10))+11
                 ENDIF
              ENDIF
         CASE LASTKEY() = 19     // กดซาย
              IF OPT < 10                    // ตวัเลือกในเมนูหลัก
                 DO PRESSUP
              ELSE
                 IF (OPT-MOD(OPT,10))/10 = 1 // ขามไปเมนูทางซายหรือสุดทาย
                    OPT = ALLSUB*10+1
                 ELSE
                    OPT = (OPT-MOD(OPT,10))-9
                 ENDIF
              ENDIF
         CASE LASTKEY() = 24     // กดลง
              DO PRESSDOWN
         CASE LASTKEY() = 5      // กดขึ้น
              DO PRESSUP
         // กรณีกด ENTER จะมาเลือกประมวลผลในสวนนี้
         CASE LASTKEY() = 13     // กดเอ็นเทอร (ENTER)
              DO CASE
                 CASE OPT = 1
                      OPT = 11
                 CASE OPT = 2
                      OPT = 21
                 CASE OPT = 3
                      @ 20,5 SAY "BYE BYE BYE"
                      EXIT
                 CASE OPT=10+ARS1 .OR. OPT=20+ARS2  // กลับเมนูหลัก

                      OPT = 1
                 CASE OPT >= 11 .AND. OPT<=10+ARS1  // กระท ําตัวเลือก
                      @ 20,5 SAY "OPT "+STR(OPT)
                      INKEY(3)
                 CASE OPT >= 21 .AND. OPT<=20+ARS2  // กระท ําตัวเลือก
                      @ 20,5 SAY "OPT "+STR(OPT)
                      INKEY(3)
              ENDCASE
         ENDCASE
         // ตัง้คา TOPLMT และ BOTTOMLMT ใหม
         DO CASE
            CASE OPT <10
                 TOPLMT = 1       ; BOTTOMLMT = ARMAIN
            CASE OPT >= 11 .AND. OPT<=10+ARS1
                 TOPLMT = 11      ; BOTTOMLMT = 10+ARS1
            CASE OPT >= 21 .AND. OPT<=20+ARS2
                 TOPLMT = 21      ; BOTTOMLMT = 20+ARS2
         ENDCASE
         DO WRITEMENU           // เขียนเมนูใหม
         KEYBOARD CHR(0)
         INKEY()                // เพือ่ตัง้คาของ LASTKEY() ใหม
     ENDIF
ENDDO
//=====================================
PROCEDURE PRESSDOWN
   IF OPT = BOTTOMLMT // ถาถงึตัวลางสุดแลวยังกดลง
      OPT = TOPLMT   // ให OPT ไปที่ตัวแรก
   ELSE
      OPT = OPT + 1
   ENDIF
RETURN
PROCEDURE PRESSUP
   IF OPT = TOPLMT    // ถาถึงตัวแรกแลวยังกดขึ้น

      OPT = BOTTOMLMT // ให OPT ไปยังตัวสุดทาย
   ELSE
      OPT = OPT - 1
   ENDIF
RETURN
//=====================================
PROCEDURE WRITEMENU
  SET COLOR TO W/B
  CLS
  FOR J = 1 TO ARMAIN
      @ 2,J*10 SAY ARCM[J]    // พิมพเมนูบนจอภาพ
  NEXT
  I = 4                                 // เมนูเริ่มพิมพบรรทัดที ่4
  DO CASE
     CASE OPT >= 11 .AND. OPT <= ARS1 + 10
          FOR J = I TO I+ARS1-1
              @ J,10 SAY ARC1[J-I+1]    // พิมพเมนูบนจอภาพ
          NEXT
          TEMPAR = ARC1
     CASE OPT >= 21 .AND. OPT <= ARS2 + 20
          FOR J = I TO I+ARS2-1
              @ J,20 SAY ARC2[J-I+1]    // พิมพเมนูบนจอภาพ
          NEXT
          TEMPAR = ARC2
  ENDCASE
  SET COLOR TO B/W
  IF OPT > 10                                   // สัง่เลือกตัวเลือกยอย
     @ I+MOD(OPT,10)-1,OPT-MOD(OPT,10);
       SAY TEMPAR[MOD(OPT,10)]                  // พิมพตัวเลือกปจจุบัน
     SET COLOR TO W/N
     @ 2,(OPT-MOD(OPT,10)) ;                    // พมิพหัวของเมนูยอย
       SAY ARCM[(OPT-MOD(OPT,10))/10]
  ELSE                                          // ขณะเลือกเมนูหลัก

     @ 2,MOD(OPT,10)*10 SAY ARCM[MOD(OPT,10)]   // พิมพหัวของเมนูยอย
  ENDIF
  SET COLOR TO B/W
RETURN

&  4.6 การเชื่อมแฟม 2 แฟม
การเชือ่มแฟมเปนหลักการที่ตองเขาใจในการศึกษาฐานขอมูล เพราะฐานขอมูลคือการนํ าแฟมหลายแฟม

มาใชรวมกนั ท ําใหการเก็บขอมูลไมซํ้ าซอน และสะดวกใช  เชนระบบเงินเดือน จะเก็บแฟมชื่อ และแฟมเงินเดือนไว
ตางกนั หากตองการพิมพเงินเดือนพรอมชื่อ ตองเชื่อมแฟมทั้ง  2 เขาดวยกัน

ตวัอยางโครงสราง และขอมูลจากแฟมชื่อ NAME.DBF
EMPID    EMPLNAME EMPLSURN
101 นาย ปติ      หมายตา
102 นาย สมปอง     นํ ้าฝน
103 นาย แทงทอง   อติชาติ
104 น.ส.ชมพู      บางบัว
105 น.ส.วาสนา     ฤดี
106 นาย แกวกลา  หทัยชนัน
107 น.ส.คมคาย     มัธยม
ตวัอยางโครงสราง และขอมูลจากแฟมเงินเดือน  SALARY.DBF
EMPID SALARY
101 12500
103 21000
104  7500
106  8600
107  6750
ตวัอยางโครงสราง และขอมูลจากแฟมขาย SALE.DBF
SALEID EMPID PRODUCT     PRICE QUANT
100001 101    รถเกง  60,000  1
100002 104    วิทยุ    5,000    8
100003 101    รถเข็น   1,000    6
100004 103    รถเกง  57,000   1
100005 107    เรือ    25,000   1
100006 107    ทีวี     5,600    4
100007 106    กระดาษ     100      300

4.6.1 การเชื่อมแฟมดวยค ําสั่ง SEEK
การใชค ําสั่ง SEEK คนหาขอมูล  ตองกํ าหนดใหแฟมที่ถูกคน  เปดแฟมขอมูลพรอมแฟมดรรชนีไว  โดยมี

ฟลดหลกัในแฟมดรรชนีตรงกับรหัสที่ตองการคนหา  โดยสงตัวแปรผานคํ าสั่ง  SEEK เขาไปคนหาในแฟมดรรชนี
: ตัวอยางที่ 4.46
// ตวัอยางนี้ใชแฟมชื่อและแฟมเงินเดือน โดยแฟมชื่อเปดแฟมดรรชนีไว
// ซึง่หนาที่หลักของโปรแกรมนี้คือ แสดงขอมูลทั้งหมดจากแฟมเงินเดือน
SELE 1; USE NAME INDEX INAME
SELE 2; USE SALARY  // ไมจ ําเปนตองเปดแฟมดรรชนีใหแฟมนี้
DO WHILE !EOF()
   _EMPID = EMPID
   ? EMPID,SALARY
   SELE 1
      SEEK _EMPID
      IF FOUND()
         ?? EMPLNAME,EMPLSURN
      ENDIF
   SELE 2  ;   SKIP
ENDDO
: ตัวอยางที่ 4.47
// ตวัอยางนี้จะคนหาจนพบแลว รวบรวมคาตาง ๆ น ํามาพิมพพรอม ๆ กัน
// โดยใชตวัอยางจากแฟมชื่อและแฟมเงินเดือน ตองการทุกเรคอรดจากแฟมเงินเดือน
SELE 1; USE NAME INDEX INAME
SELE 2; USE SALARY  // ไมจ ําเปนตองเปดแฟมดรรชนีใหแฟมนี้
SETPOS(5,2)
_TINCOME = 0
DO WHILE !EOF()
   _EMPID = EMPID
   _SALARY = SALARY
   SELE 1
      SEEK _EMPID
      IF FOUND()
         _EMPLNAME = EMPLNAME
         _EMPLSURN = EMPLSURN

      ELSE
         _EMPLNAME = SPACE(20)
         _EMPLSURN = SPACE(20)
      ENDIF
   SELE 2
   _TAX = _SALARY * 0.07
   _INCOME = _SALARY - _TAX
   _TINCOME = _TINCOME + _INCOME
   @ ROW()+1,2 SAY _EMPID
   @ ROW()  ,9 SAY _EMPLNAME + _EMPLSURN
   @ ROW()  ,50 SAY STR(_SALARY) + "  " + STR(_TAX)
   @ ROW()  ,70 SAY _INCOME
   SKIP
ENDDO
@ ROW()+1,70 SAY _TINCOME

4.6.2 การเชื่อมแฟมดวยค ําสั่ง LOCATE
การใชค ําสัง่ LOCATE คนหาขอมูล ไมจํ าเปนตองเปดแฟมดรรชนีเหมือนคํ าสั่ง SEEK เพราะคํ าสั่งนี้จะวิ่ง

เขาไปเปรยีบเทยีบกับขอมูลทีละตัวจนกวาจะพบ ท ําใหผลการทํ างานชากวาการคนหาขอมูลจากแฟมดรรชนี แตการ
ใช LOCATE จะมีความยืดหยุนมากเพราะคนหาในฟลดตาง ๆ ไดงายและเขียนไดในหลาย ๆ รูปแบบ
: ตัวอยางที่ 4.48
// ตวัอยางนี้จะคนหาจนพบแลว รวบรวมคาตาง ๆ น ํามาพิมพพรอม ๆ กัน
// โดยใชตวัอยางจากแฟมชื่อและแฟมขาย ตองการทุกเรคอรดจากแฟมขาย
SELE 1; USE NAME
SELE 2; USE SALE
DO WHILE !EOF()
   _EMPID = EMPID
   ? SALEID,EMPID,PRODUCT,PRICE,QUANT
   SELE 1  // แฟมชื่อ
      LOCATE FOR EMPID = _EMPID
      IF FOUND()
         ?? EMPLNAME,EMPLSURN
      ENDIF
   SELE 2  ;  SKIP     // แฟมขาย
ENDDO

: ตัวอยางที่ 4.49
// ตวัอยางนี ้แสดงใหเห็นการใชคํ าสั่ง LOCATE REST
// โดยระบรุหสัพนักงาน แลวคนหาในแฟมขายวาผลงานการขายของพนักงานมีอะไรบาง
// ตวัอยางนีจ้ะใชแฟมขายแฟมเดียว แลวคนหารหัสพนักงานขายจนกวาจะไมพบ
_EMPID = 0
@ 5,5 SAY "รหัสพนักงาน : " GET _EMPID
READ
USE SALE
LOCATE FOR EMPID = _EMPID
DO WHILE FOUND()
   ? SALEID,EMPID,PRODUCT,PRICE,QUANT
   SKIP
   LOCATE REST FOR EMPID =_EMPID
ENDDO
: ตัวอยางที่ 4.50
// ตวัอยางนีต้องการรายงานขอมูลการขายของพนักงานทุกคน
// โดยรายงานวาพนักงานแตละคนมีสถิติการขายเปนอยางไร
// จงึใชแฟมหลักเปนแฟมชื่อ ไปคนหาในแฟมขาย
SELE 1 ; USE SALE
SELE 2 ; USE NAME
DO WHILE !EOF()
   _EMPID = EMPID
   ? EMPID
   SELE 1
      LOCATE FOR EMPID = _EMPID
      DO WHILE FOUND()
         ? SALEID,PRODUCT,PRICE,QUANT
         SKIP
         LOCATE REST FOR EMPID =_EMPID
      ENDDO
   SELE 2
   SKIP
ENDDO

4.6.3 การเชื่อมแฟมดวยค ําสั่ง SET FILTER TO
ค ําสัง่นีใ้ชเลอืกขอมูลที่ตรงตามเงื่อนไข ซึ่งใชงานไดกวางมากกับระบบฐานขอมูล  แตใชรวมกับคํ าสั่ง

FOUND() ไมได  หากตองการตรวจสอบจ ํานวนเรคอรดที่ตรงกับชวงขอมูลที่ตองการ สามารถใชคํ าสั่ง COUNT
เมือ่ใชค ําสัง่เลอืกชวงของขอมูล  เรคอรดทั้งหมดที่โปรแกรมมองเห็นจะมีเฉพาะเรคอรดที่ตรงตามเงื่อนไข

จงึน ําไปใชหรอืตรวจสอบคาตาง ๆ ได  สิ่งที่ตองระวังกับคํ าสั่งนี้คือ หลังจากระบุขอบเขตของขอมูลแลวตองเลื่อนตัวชี้
ไปทีเ่รคอรดแรก ดวยคํ าสั่ง GO TOP

ค ําสัง่นีต้างกับค ําสั่ง SEEK และ LOCATE  ในวัตถุประสงค  นั่นคือค ําสั่งนี้ไมไดตรวจสอบวามีขอมูลใน
แฟมตามทีต่องการคนหาหรือไม  แตคํ าสั่งนี้ใชในการเลือกขอมูลตามชวงที่ระบุเงื่อนไข  เชนเลือกขอมูลที่เงินเดือน
มากกวา 5000 บาท  แตคํ าสั่ง SEEK และ LOCATE จะคนหาวามีเรคอรดที่ตองการหรือไม หากมีจะเลื่อนตัวชี้ไปที่เร
คอรดนั้น
: ตัวอยางที่ 4.51
// ตวัอยางนีจ้ะใชแฟมชื่อและแฟมเงินเดือน โดยน ํารหัสจากแฟมเงินเดือนไป
// คนหาในแฟมชื่อโดยใชค ําสั่ง SET FILTER TO
// ซึง่หนาที่หลักของโปรแกรมนี้คือ แสดงขอมูลทั้งหมดจากแฟมเงินเดือน
SELE 1; USE NAME
SELE 2; USE SALARY
DO WHILE !EOF()
   _EMPID = EMPID
   ? EMPID,SALARY
   SELE 1
      SET FILTER TO EMPID = _EMPID
      COUNT TO CNTREC
      GO TOP
      IF CNTREC > 0
         ?? EMPLNAME,EMPLSURN
      ENDIF
   SELE 2
   SKIP
ENDDO
: ตัวอยางที่ 4.52
// ตวัอยางนี ้แสดงใหเห็นการใชคํ าสั่ง SET FILTER TO
// โดยระบรุหสัพนักงาน แลวคนหาในแฟมขายวาผลงานการขายของพนักงานมีอะไรบาง
// ตวัอยางนีจ้ะใชแฟมขายแฟมเดียว แลวคนหารหัสพนักงานขายจนกวาจะไมพบ

_EMPID = 0
@ 5,5 SAY "รหัสพนักงาน : " GET _EMPID
READ
USE SALE
SET FILTER TO EMPID = _EMPID
GO TOP
DO WHILE !EOF()
   ? SALEID,EMPID,PRODUCT,PRICE,QUANT
   SKIP
ENDDO
: ตัวอยางที่ 4.53
// ตวัอยางนีต้องการรายงานขอมูลการขายของพนักงานทุกคน
// โดยรายงานวาพนักงานแตละคนมีสถิติการขายเปนอยางไร
// จงึใชแฟมหลักเปนแฟมชื่อ ไปคนหาในแฟมขาย
// ตวัอยางนีจ้ะเลือกขอมูลการขายโดยใชค ําสั่ง SET FILTER
SELE 1 ; USE SALE
SELE 2 ; USE NAME
DO WHILE !EOF()
   _EMPID = EMPID
   ? EMPID
   SELE 1
      SET FILTER TO EMPID = _EMPID
      COUNT TO CNTREC   // เกบ็จ ํานวนเรคอรดที่อยูในกลุม
      GO TOP
      FOR I = 1 TO CNTREC
         ? I,SALEID,PRODUCT,PRICE,QUANT
      NEXT
   SELE 2
   SKIP
ENDDO

4.6.4 การเชื่อมแฟมดวยค ําสั่ง SET RELATION
การระบคุวามสมัพันธอยางนอย 2 แฟม  ท ําใหงายตอการอางถึงขอมูลที่สัมพันธกัน โดยไมตองใชคํ าสั่ง

เชือ่มขอมลูแบบตาง ๆ   แตแฟมระบุความสัมพันธดวย ตองมีการเปดแฟมพรอมแฟมดรรชนีไวแลว และตองมีฟลด
หลกัของแฟมดรรชนีตรงกับฟลดหนึ่งในแฟมหลัก  จึงจะอางขอมูลถึงกันได
: ตัวอยางที่ 4.54
// การเชือ่มแฟมชื่อและเงินเดือนเขาดวยกัน
// โดยอานขอมูลจากแฟมเงินเดือนเปนหลัก
SELE 1 ; USE NAME INDEX INAME
SELE 2 ; USE SALARY
         SET RELATION TO EMPID INTO NAME
DO WHILE !EOF()
   ? SALEID,EMPID,NAME->EMPLNAME,NAME->EMPLSURN,PRODUCT,PRICE,QUANT
   SKIP
ENDDO
: ตัวอยางที่ 4.55
// การเชือ่มแฟมชื่อ เงินเดือน และขายเขาดวยกัน
// โดยใชแฟมขายเปนหลัก
SELE 1 ; USE NAME INDEX INAME
SELE 2 ; USE SALARY INDEX ISALARY
SELE 3 ; USE SALE
         SET RELATION TO EMPID INTO NAME,EMPID INTO SALARY
DO WHILE !EOF()
   _TAX = SALARY->SALARY * 0.07
   ? SALEID
   ?? EMPID,NAME->EMPLNAME,NAME->EMPLSURN
   ?? SALARY->SALARY , _TAX
   ?? PRODUCT,PRICE,QUANT
   SKIP
ENDDO

&  4.7 การสรางเสียง
ค ําสัง่ TONE สามารถสรางเสียงตามความถี่ เพื่อใหเกิดเสียงตาง ๆ ไดตามตองการ  สามารถสรางเสียง

เพลงตาง ๆ ได  แตลักษณะเสียงขึ้นกับลํ าโพงที่เครื่องคอมพิวเตอรใชงาน และความเขาใจในเรื่องของดนตรี ตัวโนต
และการท ํานองของเพลง
: ตัวอยางที่ 4.56
// ตวัอยางการสรางเสียงเพลงอยางงาย ๆ โดยขอใชทอนแรกของเพลงใกลรุงมาเปนตัวอยาง
// ตวัอยางนี้ใหเสียงไมดีนัก ตองการใหผูศึกษาปรับปรุงใหดีขึ้นตอไป
// ตวัโนตเหลานี ้ผูเขียนเทียบเสียงกับตัวเลขในหนังสือ "เลนดนตรีวิธีงาย"
   N1 =130.80 ;  N1X =138.60 ;   N2 =146.80 ;  N2X =155.60 ;   N3 =164.80
   N4 =174.60 ;  N4X =185.00 ;   N5 =196.00 ;  N5X =207.70
   N6 =200.00 ;  N6X =233.10 ;   N7 =246.90
   U1 =261.70 ;  U1X =277.20 ;   U2 =293.70 ;  U2X =311.10 ;   U3 =329.60
   U4 =349.20 ;  U4X =370.00 ;   U5 =392.00 ;  U5X =415.30
   U6 =440.00 ;  U6X =466.20 ;   U7 =493.90 ;  UU1 =523.30
//  เพลงพระราชนิพนธ : ใกลรุง
   ?T(N6);?T(U1);?T(U2);?T(N6);INKEY(0.5) //ไดยินเสียงแวว
   ?T(N5);?T(N4);?T(N5);?T(N5X);?T(N6);?T(N6);INKEY(0.5) //ดงัแผวมาแตไกล ๆ
   ?T(N4);?T(N5);?T(N6);?T(U1);INKEY(0.5) //ชุมชื่นฤทัย
   ?T(U2);?T(U1);?T(N6);?T(N5);INKEY(0.5) //หวานใดจะปาน
   ?T(N6);?T(U1);?T(N6);?T(N5);?T(N1);?T(N2);?T(N1);?T(N6) //ฟงเสียงบรรเลงขับเพลงประสาน
   ?T(N6);?T(U1);?T(N6);?T(N5);?T(N1);?T(N2);?T(N1);?T(N4) //จากทิพยวิมาน ประทานกลอมใจ  11

FUNCTION T
PARAM K
   DU = 9
   TONE(K,DU)
   INKEY(0.02)
   IF LASTKEY() = 27
      QUIT
   ENDIF
RETURN (K)

11 สิงขร  สอนขัน,  เลนดนตรีอยางงาย เลม 1, รุงแสงการพิมพ, กรุงเทพฯ, หนา 36

&  4.8 โปรแกรมขอสอบ
โปรแกรมขอสอบเปนโปรแกรมที่มีหลักการงาย ๆ คืออานขอมูลจากแฟมมาแสดง แลวรอรับคา   หากคาที่

รบัเขามาไมตรงกับที่ระบุไวจะไมไดคะแนน   หากตอบถูกคะแนนจะเพิ่มขึ้น ท ําใหโปรแกรมนี้นาสนใจในการศึกษา
หรอืเพิม่สวนเก็บประวัติของผูท ําขอสอบไว และมีการเฉลยทุกครั้งที่ทํ าเสร็จ   บางครั้งอาจเพิ่มสวนของการสุมขอสอบ
ไมใหการท ําขอสอบในแตละครั้งออกขอที่ซํ้ า ๆ กัน ทาทายยิ่งขึ้น

4.8.1 โปรแกรมขอสอบธรรมดา
แฟมคํ าถามแบบที ่1   QUES1.DBF
1. ค ําถาม QUES          X(20)
2. ตัวเลือกที่ 1 CHOICE1  X(20)
3. ตัวเลือกที่ 2 CHOICE2  X(20)
4. ตัวเลือกที่ 3 CHOICE3  X(20)
5. คํ าตอบ ANS          X(1)

: ตัวอยางที่ 4.57
// อานขอมูลจากแฟมมาแสดงทีละเรคอรด หากตอบถูกจะไดคะแนนเพิ่ม 1 คะแนน
   USE QUES1
   OK = 0
   DO WHILE !EOF()
      _ANS = " "
      ? "คํ าถาม",QUES
      ? "1.",CHOICE1
      ? "2.",CHOICE2
      ? "3.",CHOICE3
      ACCEPT TO _ANS
      IF ANS = _ANS
         OK++
      ENDIF
      ? "คะแนนของทานคือ ",OK
      SKIP
   ENDDO

: ตัวอยางที่ 4.58
// อานขอมูลจากแฟมมาแสดงทีละเรคอรด หากตอบถูกจะไดคะแนนเพิ่ม 1 คะแนน
// โดยแสดงขอมูลดวย SAY
   USE QUES1
   OK = 0
   DO WHILE !EOF()
      CLS  ;     _ANS = " "
      @ 5,5 SAY "ค ําถาม" + QUES
      @ 6,5 SAY "1." + CHOICE1
      @ 7,5 SAY "2." + CHOICE2
      @ 8,5 SAY "3." + CHOICE3
      @ 9,5 SAY "ทานเลือก : " GET _ANS
      READ
      IF ANS = _ANS   ;     OK++ ;    TONE(200,5)   ;    ENDIF
      @ 10,5 SAY "คะแนนของทานคือ " + STR(OK)
      SKIP   ;      INKEY(5)
   ENDDO

4.8.2 โปรแกรมขอสอบสุม
ในสวนนีย้งัใชแฟม QUES1.DBF เหมือนเดิม แตมีการสุมตัวเลขจากวินาทีมาใช  ท ําใหทุกครั้งที่ทํ าขอสอบ

จะขึน้ค ําถามไมซํ้ ากัน ทาทายยิ่งขึ้น
: ตัวอยางที่ 4.59
// ตวัอยางนีเ้ปนการสุมค ําถาม จากคํ าถามทั้งหมด โดยท ําไปจนหมดทุกขอ
// บรรทัดสุดทายสรุปใหรูวาท ําไดกี่เปอรเซนต
   USE QUES1
   ARNUM = ARRAY(RECCOUNT())
   FOR I = 1 TO RECCOUNT()  // สวนของการสุมคา
      RND = MOD(SECOND() * 100,RECCOUNT())+1
      IF ASCAN(ARNUM,RND) = 0
         ARNUM[I] = RND
      ELSE
         I--
      ENDIF
   NEXT

   OK = 0
   FOR I = 1 TO RECCOUNT()  // สวนของการตั้งคํ าถามใหตอบ
      GO ARNUM[I]
      _ANS = " "
      ? "คํ าถาม",QUES
      ? "1.",CHOICE1
      ? "2.",CHOICE2
      ? "3.",CHOICE3
      ACCEPT TO _ANS
      IF ANS = _ANS
         OK++
      ENDIF
      ? "คะแนนของทานคือ ",OK
   NEXT
   ? "ทานท ําไดทั้งหมด ",OK/RECCOUNT()*100," เปอรเซนต"
: ตัวอยางที่ 4.60
// ตวัอยางนีจ้ะเลือกขอสอบมาเพียง 5 ขอจากคํ าถามทั้งหมด
// สงเสียงเมื่อทํ าขอสอบถูกตอง
   USE QUES1
   ALLQUEST = 5 // จ ํานวนขอสอบที่ตองการทํ า
   ARNUM = ARRAY(ALLQUEST)
   I = 0
   DO WHILE I < ALLQUEST    // สวนของการสุมคา
      RND = MOD(SECOND() * 100,RECCOUNT())+1
      IF ASCAN(ARNUM,RND) = 0
         I++
         ARNUM[I] = RND
      ENDIF
   ENDDO
   OK := I := 0
   DO WHILE I < ALLQUEST
      CLS ; I++
      GO ARNUM[I]

PDF created with FinePrint pdfFactory Pro trial version http://www.pdffactory.com

http://www.pdffactory.com


                                                    บทที่ 4  :  หลักการพื้นฐานในการเขียนโปรแกรม                                              219

      _ANS = " "
      @ 5,5 SAY "ค ําถามขอที่ " + STR(I) + ". " + QUES
      @ 6,5 SAY "1." + CHOICE1
      @ 7,5 SAY "2." + CHOICE2
      @ 8,5 SAY "3." + CHOICE3
      @ 9,5 SAY "ทานเลือก : " GET _ANS
      READ
      IF ANS = _ANS
         OK++ ;    TONE(200,5)
      ENDIF
      @ 10,5 SAY "คะแนนของทานคือ " + STR(OK)
      INKEY(5)
   ENDDO

4.8.3 โปรแกรมขอสอบมีเฉลยเมื่อทํ าเสร็จ
เมือ่ท ําขอสอบเสร็จแลวในแตละขอยอมไดผล 2 ประการคือ ถูกหรือผิด ไมวาจะไดผลอยางไร ผูท ําขอสอบ

ยอมตองการทราบเหตุผล หรือค ําเฉลย ในหัวขอนี้จึงแสดงวิธีการเขียนโปรแกรมโดยเฉลยขอสอบเมื่อทํ าเสร็จแลว
แฟมคํ าถามแบบที่ 2   QUES2.DBF
1. ค ําถาม QUES          X(20)
2. ตัวเลือกที่ 1 CHOICE1  X(20)
3. ตัวเลือกที่ 2 CHOICE2  X(20)
4. ตัวเลือกที่ 3 CHOICE3  X(20)
5. คํ าตอบ ANS          X(1)
6. เฉลยบรรทัดที ่1 EXPLAIN1     X(20)
7. เฉลยบรรทัดที ่2 EXPLAIN2     X(20)
8. เฉลยบรรทัดที ่3 EXPLAIN3 X(20)

: ตัวอยางที่ 4.61
// แสดงสวนเฉลยอยางงาย เมื่อทํ าขอสอบเสร็จแลวในแตละขอ
   USE QUES2
   OK = 0
   DO WHILE !EOF()
      _ANS = " "
      ? "คํ าถาม",QUES

      ? "1.",CHOICE1
      ? "2.",CHOICE2
      ? "3.",CHOICE3
      ACCEPT TO _ANS
      IF ANS = _ANS
         ? "ทานตอบถูก"
         OK++
      ENDIF
      ? "คะแนนของทานคือ ",OK
      ? EXPLAIN1     // สวนที่พิมพค ําเฉลยบนจอภาพ
      ? EXPLAIN2
      ? EXPLAIN3
      SKIP
   ENDDO
: ตัวอยางที่ 4.62
// แสดงสวนเฉลยในอีกรูปแบบหนึ่ง และเลือกตัวเลือกแบบ PULLDOWN CHOICE
   USE QUES2
   OK = 0
   DO WHILE !EOF()
      CLS
      _ANS = 0
      @ 5,5 SAY "ค ําถาม" + QUES
      @ 6,5 PROMPT "1." + CHOICE1
      @ 7,5 PROMPT "2." + CHOICE2
      @ 8,5 PROMPT "3." + CHOICE3
      MENU TO _ANS    // คาที่ไดจะเปนตัวเลข
      IF ANS = LTRIM(STR(_ANS))
         OK++ ;    TONE(200,5)
      ENDIF
      @ 10,5 SAY "คะแนนของทานคือ " + STR(OK)
      @ 13,5 SAY "คํ าเฉลย :"
      @ 14,5 SAY EXPLAIN1
      @ 15,5 SAY EXPLAIN2
      @ 16,5 SAY EXPLAIN3
      SKIP  ;      INKEY(5)
   ENDDO

4.8.4 โปรแกรมขอสอบเก็บประวัติ
บอยครัง้ทีผู่ท ําขอสอบ ตองการเก็บผลการทํ าขอสอบของตน   เพื่อดูความกาวหนา หรือตองการแขงขันกับ

เพือ่นหลาย ๆ คน เมื่อทํ าขอสอบเสร็จแลวโปรแกรมเก็บขอมูลตาง ๆ เชนทํ าเมื่อใด โดยใคร และไดคะแนนเทาใด
แฟมประวัติ    HIST.DBF
1. วันที่ท ํา QDATE   X(8)
2. เวลาที่ทํ า QTIME  X(8)
3. ท ําโดย NAMET  X(20)
4. คะแนนที่ได SCORE  99

: ตัวอยางที่ 4.63
// ตวัอยางนี้สามารถเก็บขอมูลการทํ าขอสอบของผูท ําขอสอบ
// เมือ่เริม่โปรแกรมจะถามชื่อผูท ํา ตอนจบจะเก็บขอมูลโดยอัตโนมัติ
// โดยไมสนใจวาจะไดคะแนนเทาใด หรือเต็มใจที่จะบันทึกหรือไม
   _NAMET = SPACE(20)
   @ 5,5 SAY "ชื่อของผูท ําขอสอบ : " GET _NAMET
   READ
   _QDATE = DTOC(DATE())    ;   _QTIME = TIME()
   USE QUES1
   OK = 0
   DO WHILE !EOF()
      _ANS = " "
      ? "คํ าถาม",QUES
      ? "1.",CHOICE1
      ? "2.",CHOICE2
      ? "3.",CHOICE3
      ACCEPT TO _ANS
      IF ANS = _ANS  ;   ? "ทานตอบถูก"   ;    OK++   ;      ENDIF
      ? "คะแนนของทานคือ ",OK
      SKIP
   ENDDO
   USE HIST
   APPEND BLANK
   REPLACE QDATE WITH _QDATE,QTIME WITH _QTIME,;
           NAMET WITH _NAMET,SCORE WITH OK
   CLOSE ALL

&  4.9 การปรับปรุงขอมูล
การปรบัปรงุขอมูลเปนหลักการที่ส ําคัญอยางยิ่ง ในการเขียนโปรแกรมเพื่อจัดการกับฐานขอมูล เพื่อใหขอ

มลูมคีวามถกูตองและเปนปจจุบันอยูตลอดเวลา  ท ําใหนํ าขอมูลไปใชไดเหมาะสมกับเวลาและสถานที่ ... ที่สุด
การปรบัปรงุขอมลูในตอนนี้จะไมน ําตัวอยางที่ซับซอนมาแสดง ผูอานศึกษาเพิ่มเติมไดจากการปรับปรุงขอ

มลูในสวนของกรณีศึกษา ทั้ง 2 เรื่อง  ซึ่งมีวิธีการหลาย ๆ อยางในการปรับปรุงขอมูล

4.9.1 การเพิ่มขอมูลในแฟม
โดยปกตกิารเพิม่ขอมูลจะท ําการเก็บคาตาง ๆ ในตัวแปร  เมื่อพอใจแลวจะทํ าการเขียนทับลงไปบนเรคอรด

ใหมที่เปนเรคอรดวางเปลา
: ตัวอยางที่ 4.64
// ถาแฟมนี้ชื่อ FILE1 และตองการเพิ่มขอมูลเพียง 1 เรคอรดเทานั้น
   USE FILE1
   _FLD1 = 0
   _FLD2 = SPACE(20)
   @ 5,5 GET _FLD1
   @ 6,5 GET _FLD2
   READ
   APPEND BLANK
   REPLACE FLD1 WITH _FLD1, ;
           FLD2 WITH _FLD2
   CLOSE ALL
เปลา
: ตัวอยางที่ 4.65
// ถาแฟมนี้ชื่อ FILE1 และตองการเพิ่มขอมูลเพียง 2 เรคอรด ที่มีขอมูลซํ้ ากัน
   USE FILE1
   _FLD1 = 123
   _FLD2 = “ABC”
   APPEND BLANK
   REPLACE FLD1 WITH _FLD1, FLD2 WITH _FLD2
   APPEND BLANK
   REPLACE FLD1 WITH _FLD1, FLD2 WITH _FLD2
   CLOSE ALL

4.9.2 การลบขอมูล
การลบขอมูลคือ การระบุรหัสที่ตองการลบ แลวน ําไปคนหาในแฟม หากพบจะลบขอมูล

: ตัวอยางที่ 4.66
// ตวัอยางนี้ลบไดเพียงเรคอรดเดียว
   _FLD1 = 0
   @ 5,5 SAY "ระบุขอมูลที่ทานตองการลบ " GET _FLD1
   READ
   USE FILE1
   LOCATE FOR FLD1 = _FLD1
   IF FOUND()
      DELETE ;        PACK
   ENDIF
: ตัวอยางที่ 4.67
// ตวัอยางนี้ลบเปนกลุมตามเงื่อนไข
   _FLD1 = 0
   @ 5,5 SAY "ระบุกลุมขอมูลที่ทานตองการลบ " GET _FLD1
   READ
   USE FILE1
   DELETE  ALL  FOR  E STATUS  = _FLD1
   PACK

4.9.3 การแกไขขอมูล
การแกไขคอื การระบุรหัสที่ตองการลบ แลวน ําไปคนหา หากเจอจะนํ าคาในเรคอรดทั้งหมดมาเก็บในตัว

แปร  แลวรอรับการปรับปรุงตัวแปร  เมื่อปรับปรุงแลวจะเขียนทับลงไปในเรคอรดเดิม
: ตัวอยางที่ 4.68
// ตวัอยางนี้แกไขขอมูลไดเพียงเรคอรดเดียว
   _FLD1 = 0
   @ 5,5 SAY "ระบุขอมูลที่ทานตองการแกไข " GET _FLD1
   READ
   USE FILE1
   LOCATE FOR FLD1 = _FLD1
   IF FOUND()
      _FLD2 = FLD2
      @ 7,5 SAY "แกไข FLD1 " GET _FLD1
      @ 8,5 SAY "แกไข FLD2 " GET _FLD2

      READ
      IF UPDATED()
         REPLACE FLD1 WITH _FLD1 , FLD2 WITH _FLD2
      ENDIF
   ENDIF
   CLOSE ALL
: ตัวอยางที่ 4.69
// ตวัอยางนี้แกไขขอมูลทุกระเบียนดวยเงื่อนไขเดียวกัน
   _FLD1 = 0
   @ 5,5 SAY "ระบจุ ํานวนเงินเดือนที่ขึ้นใหทุกคน" GET _FLD1
   READ
   USE FILE1
   REPLACE FLD1 WITH  FLD1  +  _FLD1
   CLOSE ALL

